
علی کسایی:   نویسنده   

 )Nick Larinبرگرفته شده از دموی تکنيکی ( 

کليه حقوق این مقاله برای نویسنده و سایت طراحان ایرانی محفوظ می باشد

:مقدمه 

 به هيچ وجه غير Game Maker بعدی جذاب و خيره کننده توسط موتور موجود در نرم افزار 3ساخت بازی های 
 بعدی زیبا را 3 این نرم افزار ٬ بخوبی نيازهای اوليه ساخت یک بازی 6.0توابع موجود در نسخه ... ممکن نيست

 داخلی که بازیساز 3d Collision و culling اوليه گرفته تا الگوریتم های Projectionاز توابعی ... فراهم می آورند
ی را  درگير شدن در مفاهيم پيچيده این مباحث باز ميدارند و در عين حال کارایی قابل قبولعلاقمند و تازه کار را از
ی تکنيکی که تا دقایقی دیگر ساخت آنرا شروع خواهيم کرد ٬ سعی شده تا از اکثر در دمو... نيز ارایه می دهند

 دلار صرف خرید 20ته شود که مبلغ بهتر است فرض بر این گذاش... توابع موجود در ساختار نرم افزار استفاده شود

این نرم افزار شده است و حال ميخواهيم حداکثر بهره برداری ممکن از هزینه را داشته باشيم 

پخش یک ... همانند گذشته پيشنهاد ميکنم که این مقاله را هنگامی که درگيری ذهنی خاصی ندارید مطالعه کنيد
.يز در درک بهتر مقاله مفيد خواهد بود یک فنجان چای ن نوشيدنموسيقی ملایم و

 های بازی را در نرم افزار وارد ميکنيم و مختصات آنها را با توجه به اعداد داده شده در Spriteمطابق معمول ابتدا 
دی فایلها ا ها هنگام ورود به نرم افزار ٬ همان نام عSpriteتوجه داشته باشيد که نام ... جدول زیر تنظيم ميکنيم

...ميباشد



توضيحاتX OriginY OriginTransparentنام تصویر

wall1164True
wall2416True

spr_camera88True

spr_cam_col00False
spr_flame_light3631False

floor00False
spr_smoke1515True

spr_system00True
ceiling00False

tex_floor6363False
tex_wall00True

tex_corner6363False
spr_racer16332True
spr_racer1t6432True
spr_racer1t1615True
spr_spd_bg00True

spr_spd00True

این جلوه ... در آن وارد می کنيمرا نيز snd_thrusterبعد از اتمام مرحله بالا ٬ تنها جلوه صوتی بازی تحت نام 
... ما را خواهد داشتRacer موتور یصوتی نقش صدا

 های موجود در این دموی تکنيکی به Objectتمامی ... حال نوبت به طراحی عناصر موجود در محيط بازی ميرسد
ت موجود در این جدول یکی یکی ایجاد آنها را مطابق با اطلاعا... همراه کليه مشخصات در جدول زیر آورده شده اند

:ميکنيم 

...گذاری عناصر دقت کنيدبه حروف کوچک و بزرگ در نام : توجه* 

ObjectSpriteVisibleSolidDepthParentMaskنام 

obj_globalsspr_systemFalseFalse- 9999

cameraspr_cameraTrueTrue1spr_cam_col
obj_playerspr_racer1TrueFalse- 1spr_camera

ceilingceilingTrueFalse0
wall1wall1TrueTrue0obj_solid

wall3wall3TrueTrue0obj_solid
obj_2ddrawerspr_systemTrueFalse- 9

obj_wallfloorTrueTrue0obj_solid
obj_smokespr_smokeTrueFalse- 1

obj_solidwall1TrueTrue0
obj_road_hortex_floorTrueFalse0



obj_road_vertex_floorTrueFalse0

obj_road_c_down_lefttex_cornerTrueFalse0
obj_road_c_down_righttex_cornerTrueFalse0

obj_road_c_up_lefttex_cornerTrueFalse0
obj_road_c_up_righttex_cornerTrueFalse0

 های هميشگی Eventکليه حالات مطرح شده ٬ همان ... حال به برنامه نویسی برای عناصر فوق می پردازیم
شکلی داشتيد و یا در  ها مEventچنانچه جایی در ایجاد این ... هستند که در مقالات قبلی بارها استفاده شدند

...درک کاربرد آنها با مشکل مواجه شدید ٬ نگاهی به مقالات قبلی بياندازید

obj_globalsکدهای مربوط به 

 :Createرویداد 

//PLayer
global.player_spd = 0;
//Cars
global.car = 1;
global.car_maxspd[1] = 8;
global.car_accel[1] = 0.2;
global.car_turn[1] = 4;

cameraکدهای مربوط به 

 : Createرویداد 

d3d_start();
//sets the distant that the objects are visable at
d3d_set_fog(true,c_black,50,1000);
zoom=40;
x = obj_player.x-1;
y = obj_player.y;

 :Stepرویداد 

cdir = point_direction(obj_player.x,obj_player.y,x,y);
speed = zoom + obj_player.speed*1;
direction = obj_player.dir;
x1=obj_player.x+hspeed;
y1=obj_player.y+vspeed;
direction = cdir;
x = obj_player.x+hspeed;
y = obj_player.y+vspeed;
x+= (x-x1)/8;
y+= (y-y1)/8; 
cdir = direction;
//collision
//move_outside_solid(cdir,30);



 :Drawرویداد 

projection();

obj_playerکدهای مربوط به 

 :Createرویداد 

dir = 0;
friction = 0.1;
spd=0;
zpos = 0;
zspeed = 0;
zup = true;

 :Stepرویداد 

instance_create(self.x+4-random(8),self.y+4-random(8),obj_smoke);
if speed > global.car_maxspd[global.car]
speed = global.car_maxspd[global.car];
if zup = true zspeed+=0.01; else zspeed+=-0.01;
zpos+=zspeed;
if zspeed > 0.1 zspeed = 0.1;
if zspeed < -0.1 zspeed = -0.1;
if zpos > 0.5 zup = false;
if zpos < -0.5 zup = true;
if keyboard_check(vk_up)
{
if not sound_isplaying(snd_thruster) 
sound_loop(snd_thruster);
}
if not keyboard_check(vk_up) sound_stop(snd_thruster);

 :obj_solidرویداد برخورد با 

move_bounce_solid(false);
speed = speed/1.5;

 :Keyboard Key Leftرویداد 

dir+=global.car_turn[global.car];
if dir > (360-global.car_turn[global.car]) dir = 0;

 :Keyboard Key Rightرویداد 

dir+=-global.car_turn[global.car];
if dir < global.car_turn[global.car] dir = 360;

 :Keyboard Key Upرویداد 



motion_set(direction+dir,speed+global.car_accel[global.car]);

 :Drawرویداد 

dir2 = point_direction(y,x,camera.y,camera.x);
image_single = (dir2+dir)/(360/(image_number));
if image_single > image_number
image_single = image_single - image_number;
//FLAME LIGHT DRAW
if keyboard_check(vk_up)

{
d3d_transform_set_identity();
d3d_transform_set_scaling(1+random(0.2)+zpos/4,0.4,0.4);
d3d_transform_add_rotation_z(dir);
d3d_transform_add_translation(x,y,1);
draw_set_blend_mode(bm_add);
draw_sprite(spr_flame_light,-1,0,0);
draw_set_blend_mode(bm_normal);
d3d_transform_set_identity();

}
//SHADOW DRAW
{
d3d_transform_set_identity();
d3d_transform_set_scaling(0.6,0.6,0.6);
d3d_transform_add_rotation_z(dir);
d3d_transform_add_translation(x,y,1);
draw_sprite_ext(spr_shadow1,-1,0,0,1,1,0,0,0.4);
d3d_transform_set_identity();

}
//CAR DRAW
{
d3d_transform_set_identity();
d3d_transform_set_scaling(0.15,0.15,0.15);
d3d_transform_add_rotation_y(dir2);
d3d_transform_add_rotation_x(90);
d3d_transform_add_translation(x,y,6+zpos);
if keyboard_check(vk_up) draw_sprite(spr_racer1t,-1,0,0);
else draw_sprite(spr_racer1,-1,0,0);
d3d_transform_set_identity();

}

ceilingکدهای مربوط به 

 :Drawرویداد 

draw_set_color(c_white);
//The size and texture used for the floor
d3d_draw_floor(0,0,32,800,600,32,sprite_get_texture(ceiling,0),5,5);

wall1کدهای مربوط به 



 :Drawرویداد 

draw_set_color(c_white);
d3d_draw_wall(x-16,y,0,x+16,y,15,sprite_get_texture(tex_wall,0),1,1);

wall3کدهای مربوط به 

 :Drawرویداد 

draw_set_color(c_white);
d3d_draw_wall(x,y-16,0,x,y+16,15,sprite_get_texture(tex_wall,0),1,1);

obj_2ddrawerکدهای مربوط به 

 :Drawرویداد 

//global.car_maxspd[global.car] obj_player.speed 122

spd=(122/global.car_maxspd[global.car])*obj_player.speed + 1;

draw_set_color(c_white);
d3d_set_projection_ortho(0,0,800,600,0);
d3d_set_hidden(false);

draw_text(20,20,'A Simple Racing Demo ( Persian-Designers )
  FPS: ' + string(fps));
draw_sprite(spr_spd_bg,0,30,540);
draw_sprite_part(spr_spd,0,0,0,spd,27,33,543);
d3d_set_hidden(true);

obj_wallکدهای مربوط به 

 :Drawرویداد 

draw_set_color(c_white);
d3d_draw_wall(x-16,y,0,x+16,y,6,sprite_get_texture(tex_wall,0),1,1);

obj_smokeکدهای مربوط به 

 :Createرویداد 

image_speed = 0.25;



rot = 5-random(10);
rot2 = 0;
size = 1;
trans = 0.2;
direction = random(360);
speed = 2;
friction = 0.1;
depth = -1-random(2);

 :Animation Endرویداد 

with(self) instance_destroy();

 :Drawرویداد 

dir = point_direction(y,x,camera.y,camera.x);
rot2+=rot;
size+=0.02;
trans+=-0.0075;
{
d3d_transform_set_identity();
d3d_transform_set_scaling(0.15,0.15,0.15);
d3d_transform_add_rotation_y(dir);
d3d_transform_add_rotation_x(90);
d3d_transform_add_translation(x,y,2);  
draw_set_blend_mode(bm_add);
draw_sprite_ext(spr_smoke,-1,0,0,size,size,rot2,c_white,trans);
draw_set_blend_mode(bm_normal);
d3d_transform_set_identity();

}

obj_road_verکدهای مربوط به 

 :Createرویداد 

image_angle = 90;
y+=1;

obj_road_c_down_rightکدهای مربوط به 

 :Createرویداد 

image_angle = 90;
y+=1;

obj_road_c_up_leftکدهای مربوط به 



 :Createرویداد 

image_angle = 270;

obj_road_c_up_rightکدهای مربوط به 

 :Createرویداد 

image_angle = 180;
y+=1;
x+=1;

با این وجود هنوز یک مرحله دیگر از برنامه ... کار برنامه نویسی اشيای موجود در بازی در اینجا به پایان ميرسد
 که آنرا در projectionایجاد اسکریپت ... کدام مرحله ؟ درست حدس زدید...  باقی مانده است برنامهیسینو

بنابراین به ایجاد این تابع ميپردازیم تا آخرین مرحله برنامه ...  فراخوانی کردیمCamera مربوط به Drawرویداد 
...نویسی این دموی تکنيکی نيز به پایان برسد

:ظور کافيست تا یک اسکریپت جدید ایجاد کنيد و کدهای زیر را در آن استفاده نمایيد برای این من

projectionکدهای مربوط به اسکریپت 

with(camera) d3d_set_projection(x,y,16,obj_player.x,obj_player.y,10,0,0,1);

... رسيدایان کار برنامه نویسی این دموی تکنيکی در اینجا به پ... بسيار خوب

: برای بازی ROOMطراحی یک 

... کاری کاملا سليقه ای و دلخواه است)  برای حرکت Pathدر واقع یک ( طبيعی است که طراحی یک اتاق 
 و Game Makerبا این وجود طراحی اتاق در ...  منحصر بفردی را ایجاد کنيدROOMبنابراین هریک از شما ميتوانيد 

:موی تکنيکی نيازمند در نظر داشتن چند شرط و محدودیت است که در زیر عنوان ميشوند مخصوصا در این د

. را حتما در اتاق قرار بدهيدobj_player , camera , obj_globals , obj_2ddrawerبایست شما می)1
... استفاده شودwall1باید از ) در نمای دو بعدی طراحی اتاق ( برای دیوار های افقی )2
. استفاده شودwall3باید از ) در نمای دو بعدی ( یوار های عمودی برای د)3
را به مقداری بالاتر  ) 30( اگر از یک سيستم قدرتمند استفاده می کنيد ٬ مقدار پيش فرض سرعت اتاق )4

. افزایش دهيد تا حرکت روانتری داشته باشيد50نظير 
... ٬ همان مکان شروع بازی استobj_playerمکان قرار گيری )5
در غير اینصورت در تنظيمات .  باشد800 * 600ابعاد اتاق در اولين مرحله ساخت ٬ بهتر است بصورت )6

View تعریف کنيد و سپس رزلوشن بازی را مساوی همين مقدار 600 در 800 ابتدا یک نمای جدید با ابعاد 
. ناخواسته روی ندهدStretchقرار دهيد تا 



برای این منظور شما چهار جهت برای چهار نوع پيچ . يچ ها دقت کنيدبرای ترسيم صحيح خطوط جاده در پ)7
:ممکن را در اختيار دارید که از قرار زیرند 

obj_road_c_down_left
obj_road_c_down_right

obj_road_c_up_left
obj_road_c_up_right

 تصویر زیر به درک بهتر موضوع کمک ... فرم بالا استفاده کنيد4بسته به جهت چرخش پيچ ميبایست از یکی از 
:ميکند 

شما ميتوانيد آنرا به دلخواه خود بزرگ یا کوچک کرده ... فرم کلی اتاق در نهایت چيزی شبيه به شکل بالا ميباشد
يد  را برای دستيابی به بهترین نتيجه ممکن مجددا تنظيم کنFovميزان . و سرعت اجرای آنرا تحت کنترل در بياورید

در عين حال ميتوانيد با تغيير ... و از تکسچر های جدیدی برای تزیين کف زمين و دیواره های جاده استفاده نمایيد
یا ... نمای نشستن پشت فرمان را ایجاد کنيدپارامترهای دوربين موجود در بازی ٬ صحنه دید بازی را تغيير بدهيد و 

 و جور و کوچک برای ماشينتان طراحی کنيد تا چند آهنگ زیبا را  ٬ یک رادیوی جمعGTAاینکه به تقليد از بازی 

برایتان پخش کند  



:بازی شما بعد از اجرا چيزی شبيه تصویر زیر خواهد بود 

 از دیدن تصویر بالا چنانچه توانسته باشيد ساخت نمونه آموزشی این مقاله را با موفقيت به پایان برسانيد ٬ مسلما
با این وجود صرف کمی وقت و افزودن ظرافت و جزیيات بيشتر به کار ٬ جلوه و نمای کار شما را تا ... دخواهد برلذت 

...حد بسيار زیادی بالا می برد

ه مقایس...  دقيقه کار اضافه بر ساختار بازی می باشد30تصویری که در صفحه بعد ملاحظه می کنيد ٬ نتيجه تنها 

 عهده خودتان می گذارمایی بازی را برجزیيات و جلوه نه



...منتظر شنيدن نظرات و دیدن کارهای شما هستيم

ما و سایر دوستان در تالارهای ... چنانچه از ساخت این بازی فارغ شده اید ٬ تامل نکنيد و آنرا برای ما ارسال کنيد
...  و با هم در جهت بهبود و تکميل آنها خواهيم کوشيدسایت از شما و بازیهایتان به گرمی استقبال خواهيم کرد

...کافيست یکبار امتحان کنيد

forum/com.designers-persian.www://http:  های گفتگو و تبادل نظر  تالار

یهای کامپيوتری به زبان فارسی طراحی شده است و تا کنون سایت طراحان ایرانی با هدف آموزش ساخت باز
مدیریت سایت از . مقالات متعددی در زمينه های مختلف برنامه نویسی و ساخت بازی در آن قرار گرفته است

تمامی عزیزان علاقمند به بازی های کامپيوتری ٬ برنامه نویسان ٬ طراحان و سایر کسانی که به نحوی با بازی ها 
تباطند ٬ دعوت به همکاری به عمل می آورد تا بدینوسيله یک پایگاه علمی و موثق در زمينه صنعت ساخت در ار

.بازیهای کامپيوتری در ایران ایجاد گردد

در ضمن بسياری از نرم افزار های ساخت بازی های کامپيوتری که امروزه در سطح وسيع مورد استفاده قرار 
ده است و با مبلغ بسيار ناچيزی در اختيار علاقمندان به طراحی بازی های ميگيرند ٬ در سایت جمع آوری ش
استفاده از این نرم افزار ها در آغاز کار و به منظور آشنا شدن با اصول اوليه در . کامپيوتری قرار داده شده است

ينه ٬ به صحت موضوع طراحی بازیها بسيار موثر و مفيد بوده و شما ميتوانيد تا با چند جستجوی ساده در این زم



 را  که توسط فروشگاه الکترونيکی سایت به مشتاقان عرضه ميشوندزیر برخی از نرم افزار هاییدر. پی ببرید
 :ملاحظه می کنيد

3D Game Studio نرم افزار |Game Maker مجموعه آموزشی ساخت بازی با |مجموعه ذخایر ساخت بازی 

 بعدی3  مجموعه موتورهای سورس باز | بعدی  3یهای پکيج استاندارد ساخت باز

php?pid.index/com.designers-persian.www://http=1:    آدرس فروشگاه الکترونيکی 

 محفوظ می باشد )www.persian-designers.com ( کليه حقوق این مقاله برای نویسنده و سایت طراحان ایرانی
استفاده از مطالب موجود در این مقاله در صورت ذکر کامل منبع مجاز است


